
Zusammenfassung 

Jahreskalender werden in der Regel bei einer Agentur erstellt und dann in Druck gegeben. 
Die Agentur selbst arbeitet dann mit Programmen wie Autodesk oder Photoshop. 
Was aber wäre, wenn man solche Graphiken auch mit SAS erstellen könnte? Immerhin ist 
SAS in der Lage graphisch anspruchsvolle Reports zu generieren und ein Kalender ist im 
Wesentlichen ja auch „nur“ eine Tabelle mit Informationen. 
Nachdem es gelungen war, einen Dataset zu generieren, welcher zum Einen die Kalender-
daten und zum Anderen diese Informationen auch in der „richtigen“ Reihenfolge (Tage, 
Monate etc…) enthält, konnte man mit der Prozedur-REPORT diese in „Kalendergestalt“ 
ausgeben. Dazu wurde die seit SAS 9.4 verfügbaren Anweisungen ODS LAYOUT und 
ODS REGION verwendet. 
Das Output Delivery System (kurz ODS), welches seit SAS Version 8 zum Standard ge-
hört,  wurde in der Version 9.4 erweitert. Die Kombination aus ODS Layout und ODS Re-
gion ermöglicht es dem Anwender, verschiedene Reports, PDFs oder Sonstiges beliebig 
auf einem selbst definiertem Blanksheet via Koordinatenangabe zu platzieren. Hinter-
grundbilder und Effekte wie z.B. Transparenz können nachträglich individuell angepasst 
werden. 
Das Ziel dieses Projektes ist, dass jeder SAS Nutzer sich eigene Kalender generieren kann. 
Der dazu notwendige SAS-Code ist unter Redscope.de [1] frei verfügbar. 

1 Einleitung 
Der folgende Beitrag wird den Entstehungsprozess eines SAS-Kalenders beschreiben. 
Der Umfang erstreckt sich dabei von der Idee bis hin zu technischen Details und einem 
Ausblick auf zukünftig geplante Erweiterungen.  

1.2 Planung von Marketing-Maßnahmen 

Im September des vergangenen Jahres begann die Marketingabteilung die Planung für 
das bevorstehende Weihnachtsfest und das erste Halbjahr 2014. In diesem Zusammen-
hang wurden auch die Bestellungen von so genannten "Give-away's" wie Kugelschrei-
ber mit Firmenlogo und Jahreskalender in die Wege geleitet. 
Bei den Jahreskalendern gibt es verschiedene Möglichkeiten. Soll er möglichst funktio-
nal und kostengünstig sein? Dann bestellt man sich einen fertigen Kalender bei einer 
Online-Druckerei. Soll er möglichst individuell sein und vielleicht noch eine künstleri-
sche Note haben? Dann engagiert man eine Agentur oder eine(n) Graphikdesigner(in). 
Das wäre dann sicherlich die teuerste Variante für eine, salopp gesprochen, schön ge-
staltete "Tabelle". 

1.3 "Kalendergestalt" 

Das Stichwort "Tabelle" war dann eine schöne Gedankenbrücke zu SAS. Datasets sind 
im Grunde genommen ja nichts anderes als Tabellen mit darin enthaltenen Informatio-
nen - so auch Jahreskalender. 
Wir setzten uns daher das Ziel, ein Dataset zu erstellen, das alle notwendigen Informati-
onen eines Kalenders enthält, und anschließend in Form eines Jahresplaners auszuge-
ben. 

1.4 Planung Kalender 

Zunächst mussten dazu die Informationen eingesammelt werden. Folgende Aspekte 
sollte der Kalender abdecken: 

 Farbliche Hervorhebung von Samstagen, Sonntagen und Feiertagen 
 Anzeige der Kalenderwochen 
 Eindruck von Feiertagen 
 Platzierung von Logos in der Kopf- oder Fußzeile des Kalenders 
 Hintergrund-Bild 
 Eindruck einer Tabelle von Schulferien in die Fußzeile des Kalenders 

 
Bis wir den finalen Datensatz als Report ausgeben lassen konnten, mussten einige Vor-
arbeiten durchgeführt werden. 

2 "Step" by "Step" 
In diesem Kapitel möchte ich auf die einzelnen Schritte eingehen, die benötigt werden, 
um die SAS-Tabelle zu schaffen, in der alle benötigten Informationen vereint sind.  
Beginnen werde ich mit den gesetzlichen Feiertagen. 

2.1 Die "Osterformel" 

Eine der Anforderungen ist, dass alle gesetzlichen Feiertage in den Datumszellen ein-
getragen sind. Mein Ziel war es darüber hinaus zu gewährleisten, dass man sich optional 
ein Bundesland aussuchen kann, oder einfach alle anzeigen lässt. Deshalb sind in dem 
Datastep auch die Bundesländer spezifiziert. 
 
Nun muss jedem gesetzlichen Feiertag ein Datum zugeordnet werden. 
Die kirchlichen Feiertage wie Pfingsten und Fronleichnam können mit Hilfe der so ge-
nannten Osterformel berechnet werden. Hierbei lässt man sich über die Holiday-Funk-
tion das Datum von Ostersonntag ausgeben und addiert/subtrahiert entsprechend viele 
Tage. Andere Feiertage werden direkt angegeben (zum Beispiel der 3. Oktober oder 
Neujahr). 
 
Hier ein Code-Beispiel (Abbildung 1): 
 
Data Feiertage; 
 
(...) 
 
   Array bl BW--all; 
   dummy = Holiday('easter',&year.); 
 
   Do Over bl; bl=1;End; 
 
     Feiertag = "Neujahr           "; date = "01JAN&YEAR."d; Output; 
     Feiertag = "Karfreitag        "; date = dummy - 2     ; Output; 
     Feiertag = "Ostersonntag      "; date = dummy         ; Output; 
     Feiertag = "Ostermontag       "; date = dummy + 1     ; Output; 
 
   (...) 
 
Run; 
 
An der Stelle, wo "Do Over bl; bl=1;End;" zu lesen ist, werden die nachfolgenden Fei-
ertage allen Bundesländern zugewiesen (Abbildung 1). 

Abbildung 1: Ausschnitt aus der Tabelle Feiertage 

2.2 Wochentage 

Eine Lösung jedem Datum einen Wochentag zuzuordnen, wird im Folgenden präsen-
tiert. Diese Methode hat den Vorteil, dass man die Wochentage beliebig „Labeln“ kann, 
ohne vorher Formate definieren zu müssen (Abbildung 2). 
 
Data Wochentage; 
   Format date mmddyy8. Wochentag $40.; 
   date = Holiday('easter',&year); 
   Do While (date <= mdy(1,31,&year+1));   
      Wochentag = "Sonntag"; Output; 
      date + 7; 
   End; 
   date = Holiday('easter',&year)+1; 
   Do While (date <= mdy(1,31,&year+1)); 
      Wochentag = "Mo"; Output;  
   (...) 
run  
 
Auch hier wurde nach dem Prinzip der "Osterformel" vorgegangen. 

 

Abbildung 2: Ausschnitt aus der Tabelle Wochentage 

2.3 Notizen 

Im privatem sowie im Arbeitsumfeld wird es Termine geben, die schon lange feststehen 
und folglich auch schon mit eingedruckt werden können. Solche Termine sind bei-
spielsweise Geburtstage, Jahrestage, Jahreshauptversammlung, Quartalstreffen u.v.a. 
Diesem Zweck dient der nachstehende Datenschritt. 
 
Data mydates; 
   Length note $100; 
   date = &easter - 3; note = "Gründonnerstag"; Putput; 
   /*date = mdy(12,02,&year); note = "Redscope ;-)"; Output;*/ 
    
Run; 
 
Die auskommentierte Zeile kann als Vorlage für Ihre eigenen Eintragungen dienen. 

2.4 Each date an observation 

Es folgt nun der wichtigste Datenschritt. Zunächst wird jedem Tag im Jahr (plus dem 
Dezember der Vorjahres und dem Januar des Folgejahres) eine Observation zugewie-
sen. 
 
Data calendar_&year(drop=fdoy); 
   Format date mmddyy8. /*dow dow.*/; 
    
   Do date=mdy(12,1,&year-1) To mdy(1,31,&year+1); 
      year = Year(date); 
      monthnum = Month(date); 
      dom = Input(Put(date,day2.), 2.); 
      fdoy=Mdy(1,1,&year); 
      Length note $100; 
      note = ' '; 
      Output; 
   End;^ 
 
Run;   
 
Und schließlich werden alle bisher erzeugten SAS-Tabellen mit Hilfe der Anweisung 
MERGE verbunden: 
 
Data calendar_&year; 
   Merge calendar_&year mydates cFeiertag(keep=date Feiertag) 
      Wochentage; 
   By date; 
 
   /* Append note text to the day of month value */ 
   dom_notes = Put(dom,2.)||" " || Wochentag || Feiertag; 
 
Run; 

Im letzten Schritt werden den einzelnen Tagen (Observationen) noch die jeweiligen 
Monatsnamen zugewiesen: 
 
Data calendar_&year.; 
   Set calendar_&year.; 
   If year=&year Then 
      monthname = put (monthnum, nlstrmon.); 
   Else If year=&year-1 Then  
     Do; 
        monthname = "Dez"||Put(year,4.); 
        Call Symput ('vj_dez', monthname); 
     End; 
   Else Do; 
     monthname = "Jan"||Put(year,4.); 
     Call Symput ('fj_jan', monthname); 
   End; 
Run; 

Abbildung 3.1: Ausschnitt aus der Tabelle calender_&year. – Teil 1 

Abbildung 3.2: Ausschnitt aus der Tabelle calender_&year. – Teil 2 

Mit Hilfe der Variable Monthname können wir nun die Prozedur TRANSPOSE anwen-
den, um die Tabelle in die gewünschte Kalenderform zu bringen. 

Proc Transpose Data=calendar_&year Out=notes_calendar_&year; 
   By dom; 
   Id monthname; 
   Var dom_notes; 
Run; 
 

 

Abbildung 4: Ausschnitt aus der Tabelle notes_calender_&year. 

Unter den Monatsnamen stehen jetzt die jeweiligen Tage der Monate, beginnend mit 
dem Dezember des Vorjahres bis zum Januar des Folgejahres. (Abb. 4) 

3 Layout & "Useability" 
Dieses Kapitel widmet sich der Kalendergestaltung und benutzerfreundlichen Handha-
bung des SAS-Programms, denn der Code sollte möglichst einfach angepasst werden 
können. So sollte jeder, der SAS 9.4 nutzt und sich einen eigenen Kalender erzeugen 
möchte, einfach sein eigenes Logo anbringen können. 

Mit den seit SAS 9.4 verfügbaren Anweisungen ODS LAYOUT und ODS REGION 
sind solche „customized reports“ für jedermann möglich. 

3.1 Makros 

Die "möglichst universelle Anpassbarkeit" des Kalenders wird über eine Reihe von 
Makros und Makrovariablen erreicht. Hier beispielsweise das Kalenderjahr und der Be-
reich für die Feiertage : 

%Let year = 2014; 
%Let district = all; 

Erweitert werden diese in zukünftigen Versionen für  das Logo, Hintergrundbilder 
u.a.m. 

3.2 ODS Layout & ODS Region 

Mit den Anweisungen ODS LAYOUT und ODS REGION können SAS-Anwender Be-
richte, PDFs und Grafiken innerhalb eines Dokuments frei platzieren und kleine Ände-
rungen sehr einfach realisieren. 
 
Die OPTIONS-Anweisung legt die Größe des Gesamtdokuments und einige generelle 
Einstellungen: 

Options  
   Orientation = Landscape 
   Papersize = (80.6cm 60.6cm) 
   Nodate 
   Nonumber 
   Topmargin = .001cm 
   Bottommargin= .001cm 
   Leftmargin  = .001cm 
   Rightmargin = .001cm; 

Hier ist zu beachten, dass die Seitenbegrenzungen nicht genau auf dem Seitenrand lie-
gen dürfen. Mit 0.001cm haben wir allerdings einen kaum sichtbaren Restrand. 
 
Wir möchten ein PDF-Dokument erzeugen, keinen HTML Output. Daher wird der ODS 
HTML-Kanal geschlossen. Die DPI-Option der ODS PDF-Anweisung legt die ge-
wünschte Auflösung (dpi = xxx) fest. 
 
Ods Html Close; 
Ods Pdf File = "C:\Users\...\Kalender.pdf" Dpi=300; 
 
Nun können wir mit dem eigentlichen Gestalten beginnen. Mit 

Ods Layout Start 
   Width=80.6cm Height=60.6cm  
   ; 
(...) 
Ods Layout End; 

legen wir den gesamten Bereich fest, in dem die Felder platziert werden sollen. Die ein-
zelnen Felder werden dann mit ODS REGION definiert. Ein Beispiel wäre etwa: 

Ods Region  
   x=0cm  
   y=12.3cm 
   ; 
   Ods Pdf Text='^{Style [Cellwidth=80.6cm Cellheight=48.3cm  
           Backgroundimage="C:\Users\...\239880.jpg"]}'; 
Run; 

Die Koordinaten x und y geben exakt an wo sich das Feld im Layoutbereich platziert. 
Nach welchem Schema dies funktioniert wird noch gesondert erklärt. 
"ODS PDF text" ist eine Prozedur, die es möglich macht einen in PDF-Format hinter-
legten Fliestext zu generieren. Hierbei kann man optional die Größe des Feldes angeben 
und weitere Einstellungen vornehmen, die sich nur auf den erzeugten Text beziehen. 
Dies kann man sich zu Nutze machen  um ein Bild zu integrieren. Es wird ein PDF-
Textfeld generiert und als Hintergrundimage dient dann das Bild, welches einfügt wer-
den soll. Dies ist eine gute Methode ein Bild in der entsprechenden Qualität in SAS ein-
zubinden. Vorausgesetzt das Bild ist in der gleichen Auflösung abgespeichert wie die 
PDF (ODS PDF file="...") die angelegt worden ist. 
Ein weiteres Beispiel findet sich in Abbildung 5. 

3.2.1 Der Kalender als Report 

Wie in Abbildung 4 gezeigt hat die Tabelle bereits Kalendergestalt. Nur würde man sich 
das Ergebnis der Prozedur PRINT so nicht an die Wand hängen wollen. Mit der Proze-
dur REPORT hingegen kann man die Darstellung verbessern und durch zusätzliche Op-
tionen weiter optimieren. Auch hierzu ein Beispiel: 
 
Proc Report Data=notes_calendar_&year center nowd 
 
   Style(Report)=[Font=(Arial, 16pt)  
      Background=RGBAc1eaf2c8] 
 
   Style(Column)=[Font=(Arial, 16pt) Cellheight=1.48cm  
      Cellwidth=5.5cm] 
 
   Style(Header)=[Font=(Arial, 16pt) Font_weight=bold]  
      split='*'; 
 
   Label M_rz = 'März'; 
 
   Columns dom &vj_dez. Januar Februar M_rz April Mai Juni Juli 
    August September Oktober November Dezember &fj_jan.; 
 
   Define dom       / Order=Internal Noprint; 
   Define &vj_dez.  / Style=[Background=RGBA5984adC8] Width =10; 
   Define Januar    / Width=10 ; 
   Define Februar   / Width =10; 
   Define M_rz      / Width =10; 
   Define April     / Width =10; 
   Define Mai       / Width =10; 
   Define Juni      / Width =10; 
   Define Juli      / Width =10; 
   Define August    / Width =10; 
   Define September / Width =10; 
   Define Oktober   / Width =10; 
   Define November  / Width =10; 
   Define Dezember  / Width =10; 

   Define &fj_jan.  / Style=[Background=RGBA5984adC8] Width =10; 
 
Run; 
 
Auf die Syntax der Prozedur REPORT werde ich hier nicht näher eingehen. Allerdings 
möchte ich auf zwei interessante Eigenschaften hinweisen: 
 
Ab SAS 9.4 wird das Farbschemata RGB mit und ohne Alpha-Kanal unterstützt. Durch 
RGBA + Hexadezimalangabe der Farben ist es möglich Transparenz in Objekte und 
Hintergründe zu bekommen.  
 
Ein weiteres praktisches Tool ist die "Compute" Anweisung. Betrachten Sie hierzu fol-
gendes Beispiel: 
 
Compute Januar; 
   If substr(Januar, 4, 7) ="Sonntag"  
      Then Call Define("Januar", "Style",  
         "Style = [Background=RGBA5984adC8 Foreground=#de1f28]"); 
   If Substr(Januar, 4, 2) ="Sa"  
      Then Call Define("Januar", "Style",  
         "Style = [Foreground=#de1f28]"); 
Endcomp; 
 
In dieser Abfrage wird jede einzelne Zelle nach der Bedingung abgefragt. Somit besteht 
die Möglichkeit einzelne Zellen in einer Tabelle, beispielsweise farblich hervorzuheben. 
 
Der fertige Kalender ist in Abbildung 6 wiedergegeben und kann bei Redscope [1] her-
untergeladen werden. 

3.2.2. Allgemeines zu ODS Layout 

Per Default gilt, dass die zuerst definierte Region ganz "unten" liegt. Bei Überlappun-
gen sind also immer die zuletzt erzeugte Region zu sehen. Das erklärt, warum die Regi-
onen mit den Hintergrundbildern immer als erstes definiert werden müssen. 
 
Definiert man eine Region, muss angegeben werden, wo auf dem Bericht sie erscheinen 
soll. Dies geschieht über die Angabe der x- und y-Koordinaten. x und y beschreiben 
dabei die obere linke Ecke der Region. Der Punkt (0,0) ist im Layoutbereich ganz links 
oben verankert. 
In einem kanonischen Koordinatensystem ist der Layoutbereich der Vierte Quadrant 
und die Punkte können mit (x, -y) interpretiert werden. Das bedeutet, dass der X-Koor-
dinate ganz normal positiv nach rechtsorientiert angegeben wird, und dass die y Koor-
dinate positiv nach untenorientiert angegeben wird. 
 
Zur Veranschaulichung vergleiche man folgenden Code mit Abbildung 5: 

Options Orientation = Landscape 
   Papersize = (20.0cm 20.0cm) 
   Nodate 
   Nonumber 
   Topmargin   = .001cm 
   Bottommargin= .001cm 
   Leftmargin  = .001cm 
   Rightmargin = .001cm; 
 
Ods Html Close; 
Ods Pdf File = "C:\KSFE Bericht\odsRegion.pdf"; 
Ods Noproctitle; 
Ods Escapechar = "^"; 
Title; 
Footnote; 
 
Ods Layout Start 
   Width=20.0cm Height=20.0cm  
   ; 
 
   Ods Region x=1cm y=1cm 
   ; 
      Ods Pdf Text='^{style [Cellwidth=10cm Cellheight=15cm  
              Backgroundcolor=#000000]}'; 
   Run; 
 
   Ods Region x=5cm y=2cm 
   ; 
      Ods Pdf Text='^{style [Cellwidth=10cm Cellheight=15cm  
              Backgroundcolor=#5C5C5C]}'; 
   Run;  
 
   Ods Region x=9cm y=3cm 
      width=10cm 
   ; 
      Ods Pdf Text='^{Style [Cellwidth=10cm Cellheight=15cm  
              Backgroundcolor=#BDBDBD]}^S={Font_size=20pt}Diese Box 
ist in der obersten Ebene, da sie als letztes generiert worden ist 
und die linke obere Ecke ist bei (9,3) wobei die linken oberen Ecken 
der anderen Boxen bei (1,1) und (5,2) liegen.'; 
run; 
Ods Layout End; 
Ods Pdf Close; 

Abbildung 5: Beispiel für drei übereinandergelagerte „Regions“ 

4 Ausblick 
Das Kalenderprojekt ist ein offenes Projekt, welches keine festgesetzte Determination 
vorsieht. Wir freuen uns über Anregungen und Unterstützungen aus der SAS Commu-
nity. Die Osterformel wurde beispielsweise von einem Kollegen von SAS Institute be-
reitgestellt1. 
 
Folgende Punkte sollen unter anderem noch in diesem Jahr mit eingebaut werden: 
 
1. Eindruck der Kalenderwochen 
2. Individuelle Zellengestaltung (z.B. unterschiedliche Schriftgrößen) 
3. Makros für die Einbindung von Logos und Hintergrundbildern 
4. Einbettung in die Kybeidos Webshell 
 
Punkt 3 und 4 hängen unmittelbar zusammen, denn das Ziel sollte sein, dass jeder Nut-
zer sich einen individuellen Kalender auf der Webshell erzeugen kann, in dem er die 
erforderlichen Parameter in entsprechende Formularfelder eingibt. 
 
Wenn sie noch weitere Verbesserungsvorschläge haben, teilen Sie uns diese bitte mit, 
entweder ins SAS Forum Redscope unter Kalenderprojekt oder per Email an den Autor. 
 
 
                                           
1 http://www.redscope.org/calendar 

Abbildung 6: Der SAS Kalender 
 


